
Tips and Tricks from a (former) SAS Newbie. 
 

 

INTRODUCTION   
This paper demonstrates a few tips and tricks I learned in my progression from SAS Newbie to Base SAS certified 
programmer. Specifically they illustrate how to generate custom statistics for merge operations, and how to auto-
mate some common dataset clean-up operations I found myself doing repeatedly. 
 

DID MERGING WORK AS EXPECTED? 
For most of SAS Software users, merging datasets is a very common task, we combine data from different 
sources, we summarize long data and then need to merge the results together, etc. Sometimes it may seem we 
spend more time merging that doing analysis (and in reality data preparation takes a lot of time - but that is anoth-
er topic altogether). 
 
When I first started working with SAS software and merging datasets, I quickly realized that the standard output 
from the data step was not always sufficient for me to know if the merge worked as I expected it, or if I needed to 
check something in more detail. 
 
The typical output from a DATA step merge looks like: 
 

NOTE: There were 1000 observations read from the data set DATA.MAIN_201212. 
NOTE: The data set WORK.B has 1000 observations and 3 variables. 
NOTE: DATA statement used (Total process time): 
      real time           0.03 seconds 
      cpu time            0.01 seconds 

 
At first everything seems fine. 1,000 records were read from each file and 1,000 were outputted. However, In this 
case I happen to know that the first 25 records on WORK.A do not have a match on WORK.B and 25 records on 
WORK.B with no match on WORK.A, yet the standard output does not allow me to know that. The following 
PROC PRINT output show this (note that the 26th observation of WORK.A matches the HHID for the first obser-
vation of WORK.B) 
 

Work.A 
Obs hhid STATE segment 

22 200003854 NY 3 

23 200003864 NY 3 

24 200003882 NY 2 

25 200003886 NY 2 

26 200003906 NY 4 

27 200003940 NY 2 

28 200003960 VA 3 

29 200003962 NY 6 

Work.B 
Obs hhid ZIP group 

1 200003906 14221 O 

2 200003940 14227 A 

3 200003960 24541 M 

4 200003962 14127  

5 200003964 14224  

6 200004012 14445 B 

7 200004030 14215 H 

8 200004038 14120 M 

 
 
To improve on this, I developed an approach to collect and display my own merge statisticvs. The following SAS 
program illustrates the approach. 

data c; 
merge a (in=left) b (in=right) end=eof; � 
retain missleft missright; � 
by hhid; 
*collect the desired statistics; 
if left and not right then missright+1; � 
if right and not left then missleft+1; 
if eof then do; *output the statistics; � 
 put 'WARNING: There were ' missleft ' Record on left dataset not found on  the right; � 
  put 'WARNING: There were ' missright ' Records on right dataset not found on the left'; 
end; 
if left; *We want all records of left; 
drop miss: ; *don't forget to drop the statistic variables unless you need them; 

run; 

� The end= option creates a flag that SAS sets to 1 when there are no more records to read from either da-
taset. It is needed to print the accumulated statistics at the end of the data step only.  

 
� The retain statement creates the two statistic variables and initialize them to 0 (zero), these variables will re-

tain their values across iterations of the data step. 
 
� We collect the desired statistics and accumulate them into the retained variables. In SAS one can use var+1 

instead of the more common var=var+1 for other languages. 
 

� By starting the text string in the put statement with WARNING:, the message is printed green color like sys-
tem generated warning, Using NOTE: would have printed it in blue, ERROR: to print in red, and anything 
else will print in black. I personally chose WARNING as many of us consider blue messages as signs of suc-
cess, and a contrast color gets your attention. In this case an ERROR: was not proper as no matches could 
be perfectly acceptable or even expected. 

 
The output of the DATA step is show below: 
 

WARNING: There were 25 Record on left dataset not found on the right  
WARNING: There were 25 Records on right dataset not found on the left  
NOTE: There were 1000 observations read from the data set WORK.A. 
NOTE: There were 1000 observations read from the data set WORK.B. 
NOTE: The data set WORK.C has 1000 observations and 5 variables. 
NOTE: DATA statement used (Total process time): 
      real time           0.17 seconds 
      cpu time            0.04 seconds 

 
HOW TO CONVERT  .  (NULLS) TO ZERO’S EASILY. 
SAS stores missing numeric values as  . (dot)  in many cases this is extremely useful and the right thing to do. For 
example, in survey responses we would want to know that a person did not respond to a question and potentially 
exclude those records. Similarly some tests treat no answers differently than wrong answers. 
 
However, in other instances the missing values could be generated as part of merges (in SAS, SQL, etc.) and 
they really should be treated as zeros. Similarly a person who did not order something last quarter had zero or-
ders, not missing orders and we may want to count them as having zero orders. 
 
One can easily, use statements like if variable eq . then variable = 0 to convert these values, but in a dataset of 
many variables that need such conversion this is inefficient and potentially can create errors (plus who wants to 
type 100 such statements, certainly not me!). 
 
The following macro quickly performs the conversion for  either all numeric variables (use with care), or a list of  
specific variables. 

%macro null_to_zero(source, destination, variables); 
    data &destination; 
 set &source; 
 array names{*} � 
      %if &variables eq all or &variables eq %then _numeric_; 
   %else &variables;  
     ; � 
 
 do i = 1 to dim(names); � 
  if names{i} eq . then names{i} = 0; 
 end; 
 drop i; 
 run; 
%mend null_to_zero; 

 
 
� We define an array to contain the desired variables, so we can refer to them later in a loop which is more ef-

ficient than individual statements. 
 
� The %if … %then … %else logic is used to either load all numeric variables (using _numeric_) or the speci-

fied variables in the array. As in all macros this is effectively writing the SAS code that is then passed to the 
SAS processor. Please note that we need an extra ; (semicolon) to complete the array statement as the first 
2 semicolons are considered part of the macro processing and will not be part of the final code. 
 

� We use a do loop to cycle through all variables (using dim(names) as the upper bound) and convert any null 
values to zero. 
 

The following illustrates how the macro works. 
 

%null_to_zero(source=sample_data,destination=sample_adjusted); 
 

Before 
Obs checking savings loans 

1 1 3 0 

2 2 3 4 

3 1 1 0 

4 0 . 5 

5 3 0 0 
 

After 
Obs checking savings loans 

1 1 3 0 

2 2 3 4 

3 1 1 0 

4 0 0 5 

5 3 0 0 

. 
CONVERTING COUNT DATA TO LOGICAL INDICATORS. 
With a minor modification to the above macro, one can also quickly and efficiently convert count data (say for ex-
ample number of accounts, or number of children) into logical indicators (1=Yes, 0 = No).  

%macro as_logical(source, destination, variables); 
    data &destination; 
 set &source; 
 array names{*}  
      %if &variables eq all or &variables eq %then _numeric_; 
   %else &variables; ; 
 
 do i = 1 to dim(names); 
  if names{i} ge 1 then names{i} = 1; 
  else names{i} = 0; � 
 end; 
 drop i; 
 run; 
%mend as_logical;  

 
� Instead of converting nulsl to zeros we convert values greater or equal to 1 to 1, and anything else to 0 (to 

accommodate counts of zero or missing counts). 
 
The following illustrates how the macro works. 

 
%as_logical(source=sample_data,destination=sample_logical);
Before 

Obs checking savings loans 
1 1 3 0 

2 2 3 4 

3 1 1 0 

4 0 . 5 

5 3 0 0 
 

After 
Obs checking savings loans 

1 1 1 0 

2 1 1 1 

3 1 1 0 

4 0 0 1 

5 1 0 0 

 

CONCLUSION 

My primary motivation for writing this paper was giving back to the SAS community. I have learned a lot from the 
large number of papers available and likely could not have made as much progress as I have without them. I hope 
these tricks help some people and inspire them to write a paper or two. 
 


